


# OOABL PATTERNS & PRACTICES

EMEA PUG Challenge Workshop

**Mike Fechner** Consultingwerk, [mike.fechner@consultingwerk.de](mailto:mike.fechner@consultingwerk.de)

**Peter Judge** Progress Software, [pjudge@progress.com](mailto:pjudge@progress.com)

<https://github.com/4gl-fanatics>




# ABSTRACT

---

- Writing class files is easy. Writing good class files requires a new school of thought – especially for procedural developers. In this session we will discuss about some best practices for the OOABL. We'll be talking about best usage of Enums, Interfaces, Parameter classes and ways to manage dependencies the CCS way. The presenters will be showing OO patterns every developer should know and know when and when not to use them. We'll also be talking about the role of relational concepts like temp-tables and ProDatasets in a class based world.

## AGENDA

---



- OO terms and definitions
- General OOABL coding
  - Procedures
  - Garbage collector
  - Defining variables vs. passing references
  - Using includes
- Enumerations
- Value objects
- Static members
- Lab 1 – Create and use value objects, enums
- Contract types: interfaces & abstract classes
- Lab 2 - Refactor to interfaces
- Service Manager
- Lab 3 - Service manager implementation & use

# DEFINITIONS

Types	type defines the set of requests to which it can respond; includes classes, interfaces, enums
Strong typing	compile-time enforcement of rules
Member	stuff "inside" a type - methods, properties, variables, events etc
Access control	compile-time restriction on member visibility: public, protected, private
Class	type with executable code; implementation of a type
Abstract class	non-instantiable (non-runnable) class that may have executable code
Static	members loaded once per session. think GLOBAL SHARED
Interface	type with public members without implementations
Enum(eration)	strongly-typed name int64-value pairs
Object	running classes, aka instance

# ACCESS LEVELS

---

- Defines the visibility of members
- is enforced by the compiler

## Best practices

- it's easier to increase access once released than to decrease it
- `PROTECTED` is a decent default



# COMMON TERMS

---

Software Design Patterns: general reusable solution to a commonly occurring problem within a given context

- Parameter value
- Fluent Interfaces
- Factory method
- Singleton

SOLID principles

- S**ingle responsibility
- O**pen-closed
- L**iskov substitution
- I**nterface segregation
- D**ependency inversion

[http://en.wikipedia.org/wiki/Software\\_design\\_pattern](http://en.wikipedia.org/wiki/Software_design_pattern)

# INJECTION VS INVERSION



## INJECTION

- Can be complex, opaque
- Typically require external config and/or reflection
- Keeps single responsibility
  - Cleaner constructors
- Objects don't require a Service Manager or knowledge of one



## INVERSION

- Objects know what they need
- No need for external config or reflection
- Breaks single responsibility
- Simpler
  - Default constructor still valuable
- Depending on implementation, objects may need a Service Manager passed in



# PASTA PATTERNS

SPAGHETTI ... A HUGE MESS ALL TANGLED TOGETHER

LASAGNA ... TOO MANY UNNEEDED LAYERS  
ADDING NOTHING

RAVIOLI ... JUST ENOUGH LAYERING AND JUST  
RIGHT ENCAPSULATION



# GENERAL OOABL PROGRAMMING

---

Procedures    Garbage collector

Defining variables vs. passing references    Using includes

# PROCEDURES & OBJECTS

---

- Procedures can call classes; classes can call procedures
- Pass objects to procedures as parameters
- Lets you incrementally add OOABL
- Necessary for certain cases
  - Callbacks
  - AppServer event procedures
  - Session start (-p main.p)



# GARBAGE COLLECTOR

---

- Automatically deletes an instance if there are no references to it being held.
  - Same effect as DELETE OBJECT – runs any destructor
- References are held by
  - Variables, Properties, Temp-table fields
  - Event subscriptions
  - Progress.Lang.Object's NEXT-SIBLING and PREV-SIBLING excluded
  - SESSION:FIRST-OBJECT and FIRST-FORM chains excluded
- References are let go by
  - Variables going out of scope
  - ASSIGN <variable | property > = <some value, including ?>.
  - DELETE OBJECT
  - DELETE temp-table record
- LOG-MANAGER: LOG-ENTRY-TYPE = 'DynObjects.Class' shows manual and auto-deletion





# GC GOTCHAS

- Circular references
  - Event subscriptions
  - Parent-child with 2-way references
  - Weak references can help
- References held in persistent procs
- References held by static members
- Logging cannot tell where a reference is held

```
1. class HeaderRecord:
2.   define public property Children as DetailRecord extent no-undo
3.     get. set.
4.
5.   method public void AddDetail(pDetail as DetailRecord):
6.     assign extent(Children) = extent(Children) + 1
7.     Children[extent(Children)] = pDetail.
8.   end method.
9. end class.
10.
11. class DetailRecord:
12.   define public property Parent as HeaderRecord no-undo get.
13.
14.   constructor DetailRecord(pParent as HeaderRecord):
15.     this-object:Parent = pParent.
16.   end constructor.
17. end class.
18.
19. // caller
20. def var hdr as HeaderRecord.
21. def var detail as DetailRecord.
22.
23. hdr = new HeaderRecord().
24. detail = new DetailRecord(hdr).
25.
26. hdr:AddDetail(detail).
27. // now we have a circular reference and GC will never clean up.
28. // DELETE OBJECT will cause all of the DetailRecords to be cleaned up
29. delete object hdr.
```



# GC OF HANDLES

```
1. // Available since 11.6.3 in $DLC/[gui|src|tty]/OpenEdge.Core.pl
2. class OpenEdge.Core.WidgetHandle:
3.     define public property Value as handle no-undo
4.     get.
5.     private set.
6.
7.     // Indicates whether the handle will be destroyed/cleared when this object is destroyed.
8.     define public property AutoDestroy as logical no-undo get. set.
9.
10.    destructor public WidgetHandle():
11.        if AutoDestroy and valid-handle(this-object:Value) then
12.            delete object this-object:Value.
13.        end destructor.
14.
15.    constructor public WidgetHandle(input phValue as handle,
16.                                     input plAutoDestroy as logical):
17.        assign this-object:AutoDestroy = plAutoDestroy
18.        this-object:Value = phValue.
19.    end constructor.
20.
21.    // Does stuff with the handle
22. end class.
```

# INCLUDES AND CLASSES

---

- Objects can include includes
- Allow you to emulate generics (imperfectly)
  - Avoid CAST
- Still good for boilerplate code / generation

```
{get-service.i}
1.  CAST (Consultingwerk.Framework.FrameworkSettings:ServiceContainer
2.          :GetService(get-class("{1}":U)),
3.          {1})
// caller.p
1.  define variable svc as ISomething.
2.  // simple, easy to read
3.  svc = {get-service.i ISomething} .
4.  // as opposed to
5.  svc =
6.      CAST(Consultingwerk.Framework.FrameworkSettings:ServiceContainer
7.          :GetService(get-class(ISomething)),
8.          ISomething).
```



# VALUE OBJECTS



A **value object** is a small object that represents a simple entity whose equality is not based on identity: i.e. two value objects are equal when they have the same value, not necessarily being the same object.

Value objects should be immutable

[https://en.wikipedia.org/wiki/Value\\_object](https://en.wikipedia.org/wiki/Value_object)

# VALUE OBJECTS

---

```
/* calling */  
oModel:SetValue('Norah').
```

```
/* definition */  
class Example.Data.Model:  
    method public void SetValue (input pValue as character).
```

```
end class.
```

# VALUE OBJECTS

---

```
/* calling */
oModel:SetValue('Norah').
oModel:SetValue('Norah', 'x(20)').

/* definition */
class Example.Data.Model:
  method public void SetValue (input pValue as character).
  method public void SetValue (input pValue as character,
                               input pFormat as character).

end class.
```



# VALUE OBJECTS

```
/* calling */
oModel:SetValue('Norah').
oModel:SetValue('Norah', 'x(20)').
oModel:SetValue('Norah', 'x(20)', 'UTF-8').

/* definition */
class Example.Data.Model:
    method public void SetValue (input pValue as character).
    method public void SetValue (input pValue as character,
                                input pFormat as character).
    method public void SetValue (input pValue as character,
                                input pFormat as character,
                                input pEncoding as character).
end class.
```

# VALUE OBJECTS

```
/* calling */
oString = new Example.String().
oString:Value = 'Norah'.

oModel:SetValue(oString).

/* definition */
class Example.String:
    define public property Value as character no-undo get. set.

end class.

class Example.Data.Model:
    method public void SetValue (input pValue as Example.String).
end class.
```

# VALUE OBJECTS

```
/* calling */
oString = new Example.String().
oString:Value      = 'Norah'.
oString:Format     = 'x(20)'.
oString:Encoding   = 'UTF-8'.
oModel:SetValue(oString).

/* definition */
class Example.String:
  define public property Value      as character no-undo get. set.
  define public property Format     as character no-undo get. set.
  define public property Encoding   as character no-undo get. set.
end class.

class Example.Data.Model:
  method public void SetValue (input pValue as Example.String).
end class.
```



# IMMUTABLE VALUE OBJECTS

```
/* calling */
oString = new Example.String('Norah', 'x(20)', 'UTF-8').
oModel:SetValue(oString).

// definition
1. class Example.String:
2.   // read-only properties
3.   define public property Value      as character no-undo get. PRIVATE set.
4.   define public property Format     as character no-undo get. PRIVATE set.
5.   define public property Encoding   as character no-undo get. PRIVATE set.
6.   // values set in constructor(s)
7.   constructor public String(input pValue      as character,
8.                               input pFormat    as character,
9.                               input pEncoding  as character):
10.    assign this-object:Value      = pValue
11.        this-object:Format       = pFormat
12.        this-object:Encoding     = pEncoding.
13.   end constructor.
14. end class.

class Example.Data.Model:
  method public void SetValue (input pValue as Example.String).
end class.
```

# ENUMS



An **enumerated type** (also called enumeration, `enum[...]`) is a data type consisting of a set of named values called elements, members, enumeral, or enumerators of the type. The enumerator names are usually identifiers that behave as constants in the language

[https://en.wikipedia.org/wiki/Enumerated\\_type](https://en.wikipedia.org/wiki/Enumerated_type)



# WHY USE ENUMS

```
procedure SetOrderStatus(input pOrderNum as integer,  
                        input pStatus as integer):  
    OpenEdge.Core.Assert:IsPositive(pStatus, 'Order status').  
  
    find Order where Order.OrderNum eq pOrderNum exclusive-lock.  
    assign Order.OrderStatus = pStatus.  
    // what happens when the integer isn't a valid status? How do we know ?  
end procedure  
  
run SetOrderStatus (12345, 0).    // this throws an error via the Assert  
run SetOrderStatus (12345, 1).    // what is this status?  
run SetOrderStatus (12345, 2).
```

# DEFINING ENUMS

```
// enum type
enum Example.Orders.OrderStatusEnum: //implicitly FINAL and cannot be extended
  // enum member
  define enum    Shipped           = 1  // default start at 0
                  Backordered       // = 2 . Values incremented in def order
                  Ordered
                  Open
                  Cancelled         = -1 // historical set of bad values
                  UnderReview       = -2

                  Default = Ordered.    // A single value can have many names
  // enum members are the only members allowed
  // enum members can only be used in enum types
end enum.
```



# USING ENUMS

```
procedure SetOrderStatus(input pOrderNum as integer,  
                        input pStatus    as Example.Orders.OrderStatusEnum):  
    //ensures that we have a known, good status  
    OpenEdge.Core.Assert:NotNull(pStatus, 'Order status').  
  
    find Order where Order.OrderNum eq pOrderNum exclusive-lock.  
    assign Order.OrderStatus = pStatus:GetValue().  
    //Alternative way of getting the value  
    assign Order.OrderStatus = integer(pStatus).  
end procedure
```

`run SetOrderStatus (12345, OrderStatusEnum:None).`      `// COMPILE ERROR`  
`run SetOrderStatus (12345, OrderStatusEnum:Backordered).`  
`run SetOrderStatus (12345, OrderStatusEnum:Ordered).`

`run SetOrderStatus (12345, ?).`      `// this throws an error via the Assert`

# STATIC MEMBERS



A static member is scoped to the defining class type and makes it available for the duration of the session without the need to instantiate a member of that class



# USING STATIC MEMBERS

---

- Easy to use: no NEW required
- Properties, variables, methods, events, temp-tables/datasets, buffers can all be STATIC
- Once loaded, always available for the duration of the session
- Members can hold object references (instances) which may be explicitly destroyed
- Are never part of an interface
- Can never be inherited
- Prevent the class from being part of the reusable object cache

# WHAT SHOULD THEY BE USED FOR?


---

- ✓ Factories and builder methods
- ✓ Session-wide caches
- ✓ To replace GLOBAL SHARED variables
- ✓ Session-wide “anywhere” events



# LABS:WHAT ARE WE BUILDING

- <https://github.com/4gl-fanatics>
  - Before & Solutions



Customer Report Parameters

Date from: 01-Jan -19

Date to: 31-Dec-19

Customer from: 0

Customer to: 99

Output file: report.pdf

OK Cancel

The premise is a UI to capture a set of parameters for a customer report, and validate the values specified. Default values may be provided as inputs.

The parameter values are returned to a caller , together with a code indicating success or failure of the validation.

## ABL UI Dialog

The dialog box `abl/d-customerreport-parameter.w` is used to create/input and validate parameters for a customer report using ABL GUI.

To run this dialog, call `run-report.p`.

## GUI for .NET Dialog

The form `gui/CustomerReportParam.cls` is used to create/input and validate parameters for a customer report using GUI for .NET.

To run this dialog, call `run-report-net.p`.

## Inputs

The dialogs asks for a date range (from & to), a customer number range (from & to) and a output file name.

## Logic

If the initial input to-date or the to-custnum are null/unknown, then only ask for a from- value (ie range with a floor).

If the initial input from-date or the from-custnum are null/unknown, then only ask for a to- value (ie range with a ceiling).

Basic validation is also performed:

- from- values must be  $\leq$  to- values
- a filename must be specified

## Outputs

In addition to the screen values, a character result is returned: one of OK or CANCEL to indicate whether the report should be printed or not.

# LABS:WHAT ARE WE BUILDING

---

- <https://github.com/4gl-fanatics>
  - Before & Solutions

C:\workshop\dev\before  
C:\workshop\dev\after

## LAB: VALUE OBJECTS, ENUMS

---

Create and use a *value object* for the  
parameters

Create and use an *enum* for the  
result



# CONNECTION DETAILS

---

Wifi details

SSID <>

Password <>

Connecting to the VM

**\Administrator**

**PR0workshop2019**



# AFTER LAB I

---

✓ You now have

- ... a class containing the report parameter values
- ... a UI that consumes and produces those values
- ... a class to validate those values
- ... an enum to represent success or failure

- What if

- ... you want to change the number of inputs?

- Eg. a date range with only a floor (only from a value)

- ... you want to change the input control types?

- Eg ABL FILL-IN to RadDateTimePicker

# CODING TO A CONTRACT

---

Interfaces

Abstract classes

# SOFTWARE GOALS

---

- Loosen dependencies between objects
  - Easier to change/replace/extend behaviour
  - Easier to test (swap out real objects for doppelgängers)
- Extensibility
  - Need capability to add and extend object behaviour
  - May not have ability to change base behaviour (no/encrypted source code)
- Lower the impact of changes



# USE CONTRACT TYPES FOR DEFINITIONS, PARAMETERS

---

- Use them to define the programming interface
  - Compiler requires that implementing/concrete classes fulfill a contract
  - Neither can be instantiated aka NEW'd : application requires a component / code to provide a runnable class that satisfies the contract
- Interfaces preferred
  - Can use multiple at a time
  - Now have *I-won't-break* contract with implementers
- Use inheritance for common or shared behaviour
  - Careful of deep hierarchies – reduces flexibility



# CONTRACT TYPES

---

## INTERFACES

- Public contract
- Effectively a header – can never have any implementation
- Can inherit from interfaces and be extended by other interfaces
- Members must be implemented by an implementer

## ABSTRACT CLASSES

- Public or protected contract
- May have abstract members that must be implemented in a descendant
- May have concrete / implemented members (ie with behaviour etc)
- Can implement interface(s)
- Can inherit from and be extended by abstract or concrete classes




## LAB: INTERFACES AND ABSTRACT CLASSES

---

Create *interfaces* for

- Validators and other objects
- Methods that consume them

You may want to consider a class  
that knows how to build the dialog  
and pass in values



# AFTER LAB 2

---

- ✓ You now have

- ... a validator interface that's implemented by the value validator from Lab 1
- ... an interface that takes a validator, parameter value object and returns a result enum, and one or more implementations of the interface

- You may have

- ... Classes and/or interfaces that represent the UI controls you want to prompt for

# SERVICE MANAGER



The Service Manager is the infrastructure component that manages the relationship between the abstraction (an OO type name) and the implementing instance (a concrete, instantiable OO type name).

Edu, M., Fechner, M., Prinsloo S. L., Judge, P., & Smith, R. (2016). *Service Manager Specification OpenEdge Application Architecture Specification (OEAA) version 1.0*. Progress Software.

# WHAT IS THE COMMON COMPONENT SPECIFICATION (CCS) PROJECT?

---

- A project for developing standard business application component specifications for business applications, driven by OpenEdge experts and evangelists from the entire Progress Community
- Defines a common understanding and language of an business application architecture
- Enables standards-based framework components that can easily interoperate
- Enables creation of standard tools for those components

# CCS OUTPUTS

---

- Versioned specifications in document form
- Interface definitions – OOABL source code
  - Also includes many interfaces, many enums and one class
  - In 11.7.2 all published interfaces included in OE install

The behavior described in the document and interface(s) **MUST** be followed in order to claim compatibility with a version of a specification

- May include some sample code or test cases
  - No complete reference implementation as part of the project

<https://github.com/progress/CCS>



# SERVICE MANAGER RESPONSIBILITIES

---

- The Service Manager provides two categories of functionality

## 1. Service name resolution

```
/* Returns a usable instance of the requested service.  
   @param P.L.Class The service name requested  
   @return P.L.Object A usable instance  
   @throws P.L.AppError Thrown when no implementation can be found */  
method public Progress.Lang.Object getService(  
    input poService as class Progress.Lang.Class).
```

## 2. Lifecycle management

```
/* Destroys and flushes from any cache(s) objects scoped to the argument scope.  
   @param ILifecycleScope A requested scope for which to stop services. */  
method public void stopServices(  
    input poScope as Ccs.ServiceManager.ILifecycleScope).
```

# COMPLETE CCS.COMMON.ISERVICEMANAGER

---

```
interface Ccs.Common.IServiceManager inherits Ccs.Common.IManager:
    /* Returns a usable instance of the requested service.
       @param P.L.Class The service name requested
       @return P.L.Object A usable instance */
    method public Progress.Lang.Object getService(input poService as class Progress.Lang.Class).

    /* Returns a usable instance of the requested service.
       @param P.L.Class The service name requested
       @param ILifecycleScope A requested scope. The implementation may choose to ignore this value.
       @return P.L.Object A usable instance */
    method public Progress.Lang.Object getService(input poService as class Progress.Lang.Class,
                                                input poScope as Ccs.ServiceManager.ILifecycleScope).

    /* Returns a usable instance of the requested service.
       @param P.L.Class The service name requested
       @param character An alias for the service. The implementation may choose to ignore this value.
       @return P.L.Object A usable instance */
    method public Progress.Lang.Object getService(input poService as class Progress.Lang.Class,
                                                input pcAlias as character).

    /* Destroys and flushes from any cache(s) objects scoped to the argument scope.
       @param ILifecycleScope A requested scope for which to stop services. */
    method public void stopServices(input poScope as Ccs.ServiceManager.ILifecycleScope).
end interface.
```

# IMPLEMENTING A SERVICE MANAGER

---

1. How does it know what implementation to return for a requested service (type)?
  - Use a database- or temp-tables as some form of registry?
  - Pattern matching?
2. How does it know how to run (instantiate) an implementation?
  - What if there's no default constructor?
  - What if there are required arguments?
3. How long does object live?
  - An application cache of objects
4. How do you get a reference to it?
  - A service manager of service managers?



# IMPLEMENTING A SERVICE MANAGER

---

## MAPPING



1. How does it know what implementation to return for a requested service (type)?

- Use a database- or temp-tables as some form of registry?
- Pattern matching?

## PROVIDERS



2. How does it know how to run (instantiate) an implementation?

- What if there's no default constructor?
- What if there are required arguments?

## SCOPE



3. How long does object live?

- An application cache of objects

## DISCOVERY



4. How do you get a reference to it?

- A service manager of service managers?

# MAPPING: HOW DO WE KNOW WHICH IMPLEMENTATION MATCHES A SERVICE (TYPE)?

---

- Naming conventions
- Explicit
  - Flat file
  - Database
  - Code

Service (INTERFACE)	Alias	Implementation (CLASS)
4GIFanatics.UI.IInputFormHelper	GUI	4GIFanatics.UI.Implementation.AblInputFormHelper
4GIFanatics.UI.IInputFormHelper	NET	4GIFanatics.UI.Implementation.FormInputFormHelper



# INJECTION PROVIDERS

---

- Injection types: **CONSTRUCTOR** , METHOD , PROPERTY
- General purpose
  - Use the default constructor
  - Picks a constructor via reflection
  - Write once, use often
- Tailored to an implementation type
  - Knows what the dependencies are
  - Knows , and how to inject them: constructor, method, property
  - Less reuse

# PROVIDERS: SIMPLE EXAMPLE (NO DEPENDENCIES)

---

```
1. class 4GIFanatics.ServiceManager.Provider.DefaultConstructorProvider implements IProvider:
2.     /* Instantiates (NEWs) a class. */
3.     method public Progress.Lang.Object invokeService(pImplType as Progress.Lang.Class,
4.                                                     pContext as Progress.Lang.Object):
5.         define variable svc as Progress.Lang.Object no-undo.
6.         Assert:NotNull(pImplType, 'Implementation type').
7.         Assert:NotAbstract(pImplType).
8.         Assert:NotInterface(pImplType).
9.
10.        svc = pImplType:New().
11.
12.        if type-of(svc, IService) then
13.            cast(svc, IService):initialize().
14.
15.        return svc.
16.    end method.
17. end class.
```

# SCOPE: MANAGING THE IMPLEMENTATION'S LIFESPAN


---

As long as an object respects a contract – it implements an interface or inherits from an abstract class - does it matter who created it and how?

Or how long it lives for after an object is done using it?

- How long should an object live?

It could be for the life of the ...

- Session  singleton
- User's login session
- Request
- As long as it is used ("transient")

- How many instances should there be?

# DISCOVERY: CCS.COMMON.APPLICATION CLASS

---

- The only class CCS will ever publish
- Naming it *Framework* seemed wrong
- Provides access to the Startup Manager
- Provides access to the Service Manager (for convenience)
- Yes – it is like a GLOBAL SHARED variable

... but there didn't seem to be a better way

# CCS.COMMON.APPLICATION

```
CLASS Ccs.Common.Application FINAL:
  // Provides access to the injected IStartupManager.
  DEFINE STATIC PUBLIC PROPERTY StartupManager AS Ccs.Common.IStartupManager NO-UNDO GET. SET.

  // Provides access to the injected IServiceManager.
  DEFINE STATIC PUBLIC PROPERTY ServiceManager AS Ccs.Common.IServiceManager NO-UNDO GET. SET.

  // Provides access to the injected ISessionManager.
  DEFINE STATIC PUBLIC PROPERTY SessionManager AS Ccs.Common.ISessionManager NO-UNDO GET. SET.

  // Version of the Common Component Specification implementation.
  DEFINE STATIC PUBLIC PROPERTY Version AS CHARACTER NO-UNDO INITIAL '1.0.0':u
  GET.
  // Prevent creation of instances.
  CONSTRUCTOR PRIVATE Application ():
    SUPER ().
  END CONSTRUCTOR.
END CLASS.
```



# USING CCS.COMMON.APPLICATION

```
// session_start.p
using 4GIFanatics.ServiceManager.*.

// start & init the IServiceManager implementation
Ccs.Common.Application:ServiceManager = new ConfigFileSvcMgr().
Ccs.Common.Application:ServiceManager:initialize().
```

```
using 4GIFanatics.UI.*.
define variable svc as Progress.Lang.Object no-undo.
define variable ifh as IInputFormHelper no-undo.

svc = Ccs.Common.Application:ServiceManager
      :getService(get-class(IInputFormHelper)).

ifh = cast(svc, IInputFormHelper).
// use the Input Form Helper
```


## LAB: IMPLEMENT A SERVICE MANAGER

---

Create a Service Manager that implements the CCS IServiceManager interface

Assign it to the CCS Application static property

Replace all (?) NEW() calls with getService()



# WHAT DID WE LEARN?

---

- OOABL terminology
- What is garbage collection
- Using (immutable) value objects
- Enumerations
- Using interfaces & abstracts to create plug-n-play components

## USEFUL LINKS / SITES

<https://github.com/4gl-fanatics>

<https://github.com/progress/CCS>



Peter Judge  
Mike Fechner

[pjudge@progress.com](mailto:pjudge@progress.com)  
[mike.fechner@consultingwerk.de](mailto:mike.fechner@consultingwerk.de)